Load Modules

```
> # read "integral_bases.txt"
> read "HermiteReduction.mm"
> with(HermiteReduction):
```

Additive Decomposition

Calling sequence:

AdditiveDecomposition(f, basis, Dx, x, L)

Input: f, an element in C(x)[Dx]/<L>;

basis, basis= [W, deg], where W is a normalized integral basis of C(x)[Dx]/<L> and deg is a list of integers such that deg[i] = -val_infty(W[i]);

Dx, name, a differential operator with repect to x;

x, a variable name;

L, a linear differential operator in C(x)[Dx].

Output: [g, [P, W], [Q, V], h],

where g, h in C(x)[Dx]/<L>, P, Q are list of coefficients in C(x)^n, W is a global integral basis that is normal at infinity and

V is a local integral basis at infinity such that

f = g' + h with h = P * W + Q * V

and f is integrable in C(x)[Dx]/<L> if and only if h = 0.

Description: In the output V = TW with T = diag(x^(-deg[1]), ..., x^(-deg[n])), where n = dim(W).

```
> # Example 1: nonfuchsian at 0
> f1 := -1-2/ x^2 + (-2+3*x^2 -3*x^4)*Dx/x;
```

$$f1 := -1 - \frac{2}{x^2} + \frac{\left(-3\,x^4 + 3\,x^2 - 2\right)\,Dx}{x} \quad (2.1)$$

```
> L1 := x^3*Dx^2+(2+3*x^2)*Dx;
```

$$L1 := x^3\,Dx^2 + \left(3\,x^2 + 2\right)\,Dx \quad (2.2)$$

```
> #basis1 := [normalized_integral_basis(L1)];
  basis1 := [[x^3*Dx, 1], [0, 0]];
```

$$basis1 := \left[\left[x^3\,Dx, 1\right], \left[0, 0\right]\right] \quad (2.3)$$

```
> g1, P1, Q1, h1 := op(AdditiveDecomposition(f1, basis1, Dx, x, L1)
  );
```

$$g1, P1, Q1, h1 := 2\,x^2\,Dx + \frac{2}{x} - 3\,x^4\,Dx - x, \left[\left[0, 0\right], \left[x^3\,Dx, 1\right]\right], \left[\left[0, 0\right], \left[x^3\,Dx, 1\right]\right], 0 \quad (2.4)$$

```
> h1; ## f1 is integrable since its reminder h1 is zero
```

$$0 \quad (2.5)$$

```
> Verify_HR(f1, g1, h1, Dx, x, L1);
```
$$true \tag{2.6}$$

```
> # Example 2: nonfuchsian at infinity
> L2 := x*Dx^2 -(3*x^3+2)*Dx;
```
$$L2 := x\,Dx^2 - \left(3\,x^3 + 2\right)\,Dx \tag{2.7}$$

```
> f2 := 4*x^3 + 1/x*Dx;
```
$$f2 := 4\,x^3 + \frac{Dx}{x} \tag{2.8}$$

```
> #basis2 := [normalized_integral_basis(L2)];
  basis2 := [[Dx/x^2, 1], [0, 0]];
```
$$basis2 := \left[\left[\frac{Dx}{x^2}, 1\right], [0, 0]\right] \tag{2.9}$$

```
> g2, P2, Q2, h2 := op(AdditiveDecomposition(f2, basis2, Dx, x, L2)
  );
```
$$g2, P2, Q2, h2 := \frac{\left(-\frac{1}{3}\,x^4 + \frac{4}{9}\,x\right)\,Dx}{x^2} + x^4 - \frac{4\,Dx}{9\,x} + \frac{4\,x}{3} + \frac{Dx}{2} - \frac{3\,x^2}{2}, \left[[0, 0], \left[\frac{Dx}{x^2}, 1\right]\right], \left[\left[0, -\frac{4}{3} + 3\,x\right], \left[\frac{Dx}{x^2}, 1\right]\right], -\frac{4}{3} + 3\,x \tag{2.10}$$

```
> h2;## f2 is not integrable since its reminder h2 is not 0
```
$$-\frac{4}{3} + 3\,x \tag{2.11}$$

```
> Verify_HR(f2, g2, h2, Dx, x, L2);
```
$$true \tag{2.12}$$

```
> # Example 3: hyperexponential case
> f3 := 1;
```
$$f3 := 1 \tag{2.13}$$

```
> L3 := Dx - x/(x^2+1) + 2/(x-1);
```
$$L3 := Dx - \frac{x}{x^2 + 1} + \frac{2}{x - 1} \tag{2.14}$$

```
> #basis3 := [normalized_integral_basis(L3)];
  basis3 := [[(x - 1)^2], [1]];
```
$$basis3 := \left[\left[(x - 1)^2\right], [1]\right] \tag{2.15}$$

```
> g3, P3, Q3, h3 := op(AdditiveDecomposition(f3, basis3, Dx, x, L3)
  );
```

$$g3, P3, Q3, h3 := -x + 1 - \frac{(x-1)^2}{2}, \left[\left[\frac{1}{2\,(x-1)}\right], \left[(x-1)^2\right]\right], \left[\left[\frac{x}{2\,x^2+2}\right], \left[\frac{(x-1)^2}{x}\right]\right], \frac{x^3-x}{2\,x^2+2} \tag{2.16}$$

```
> Verify_HR(f3, g3, h3, Dx, x, L3);
```

$$true \tag{2.17}$$

Creative Telescoping

Calling sequence:

CreativeTelescoping(f, ann, basis, Der_x, Der_t)
CreativeTelescoping(f, ann, basis, Der_x, Der_t, 'cert')

Input: f, an operator in C(x, t)[Dx, Dt];
ann, ann = [L, P] where L in C(x,t)[Dx] and P = cDt - U_t with c in C(x,t) and U_t in C(x, t)[Dx];
Der_x, Der_x = [Dx, x], where Dx is a differential operator with repect to x and x is a variable name;
Der_t, Der_t = [Dt, t], where Dt is a differential operator with repect to t and t is a variable name.

Output: L, where L in C(t)[Dt] and g in C(x)[Dx, Dt]/ann such that
L(f) = Dx(g)
and L is of minimal order.

Optional: The optional argument is assigned cert = g.

```
> # Example 4: in van Der Hoeven paper 2021
```

```
> f4 := 1; ## f corresponds to sin(t*x)*exp(-x^2)
```

$$f4 := 1 \tag{3.1}$$

```
> ann4 := [Dx^2 + 4*x*Dx +(2+t^2+4*x^2), Dt - x/t*Dx - 2*x^2/t];
```

$$ann4 := \left[Dx^2 + 4\,x\,Dx + t^2 + 4\,x^2 + 2, Dt - \frac{x\,Dx}{t} - \frac{2\,x^2}{t}\right] \tag{3.2}$$

```
> #basis4 := [normalized_integral_basis(ann[1])];
  basis4 := [[2*x + Dx, 1], [0, 0]];
```

$$basis4 := [[2\,x + Dx, 1], [0, 0]] \tag{3.3}$$

```
> CT4:= CreativeTelescoping(f4, ann4, basis4, [Dx, x], [Dt, t],
  'g4');
```

```
check order 1
```

(3.4)

$$CT4 := \frac{t}{2} + Dt \tag{3.4}$$

```
> g4;
```

$$-\frac{2\,x + Dx}{2\,t} \tag{3.5}$$

```
> VerifyTelescoper(CT4, f4, g4, ann4, [Dx, x], [Dt, t]);
```

$$true \tag{3.6}$$

> *# Example 5: Hyperexponential case*

```
> H5 := sqrt((t-x)/x)*exp(t^2*(x-2*t));
```

$$H5 := \sqrt{\frac{t-x}{x}}\,\mathrm{e}^{t^2\,(x-2\,t)} \tag{3.7}$$

```
> L5 := Dx - simplify(diff(H5, x)/H5);
```

$$L5 := Dx - \frac{2\,t^3\,x - 2\,t^2\,x^2 - t}{2\,(t-x)\,x} \tag{3.8}$$

```
> P5 := Dt - simplify(diff(H5, t)/H5);
```

$$P5 := Dt - \frac{-12\,t^3 + 16\,t^2\,x - 4\,x^2\,t + 1}{2\,t - 2\,x} \tag{3.9}$$

```
> ann5 := [L5, P5];
```

$$ann5 := \left[Dx - \frac{2\,t^3\,x - 2\,t^2\,x^2 - t}{2\,(t-x)\,x},\, Dt - \frac{-12\,t^3 + 16\,t^2\,x - 4\,x^2\,t + 1}{2\,t - 2\,x}\right] \tag{3.10}$$

```
> # basis5 := [normalized_integral_basis(L5)];
  basis5 := [[x], [1]];
```

$$basis5 := [[x], [1]] \tag{3.11}$$

```
> f5 := 1;
```

$$f5 := 1 \tag{3.12}$$

```
> CT5 := CreativeTelescoping(f5, ann5, basis5, [Dx, x], [Dt, t],
  'g5');
```

```
check order 1
check order 2
```

$$CT5 := \frac{36\,t^6 + 45\,t^3 - 4}{2\,t^2} + \frac{\left(9\,t^3 + 2\right)\,Dt}{t} + Dt^2 \tag{3.13}$$

```
> g5;
```

$$\frac{36\,t^6+45\,t^3-4}{2\,t^4}+\frac{\left(9\,t^3+2\right)\left(\frac{2\,x}{t}+\frac{-6\,t^3-2}{t^3}\right)}{t}-\frac{x}{2\,(t-x)\,t}+4\,x^2+\frac{-24\,t^3\,x-6\,x}{t^2}+\frac{3\left(24\,t^6+5\,t^3+4\right)}{2\,t^4} \tag{3.14}$$

```
> VerifyTelescoper(CT5, f5, g5, ann5, [Dx, x], [Dt, t]);
```

$$true \tag{3.15}$$